
//*********************************************
//
// NAME: WPCHTB01.WPM
// PURPOSE: Coach on Creating a table
// AUTHOR: J. J. Westergren, WPCorp
//   Rev Date: 5/28/93
//*********************************************
// Revision History
//
//
Global(ScrnMode;MsgAdj;DefTitle;Title)
Global(FakeBtns;CancelMsg)
Global(w;h;tbox;hbox;helpmsg)
Global (HotElement;SaveScrn)
select[0]=0

DefTitle="Coach" // Main "Coach" name
Title="Coach"
ScrnMode=?ScreenHigh-24

Use("wpchlibr.wpm") // Shared library file
Use("wpchlbtb.wpm")

If(Not Exists(Module))
ScreenSetup()
ColorScheme()

Endif

If(?BlockActive)
Blockoff

Endif

OnCancel Call(CancelCoach)
Error(off!)
MsgAdj=DlgMessage()
CancelMsg=0

//*********************************************
//
// MAIN PROGRAM: Coach
//
//*********************************************
//Label (menu)
WindowMaximize
Cancel(Off!)
Success=0
If(NOT Exists(Module))

Call(OverviewDlg) // Overview Dialog
If(Result=-1)
Windowmaximize
QuitCoach()
Return

Endif
Endif

//*********************************************************
// Create a table?
//***********************************************************

Call(CreateTable)

If(NOT Exists(Module))
ExitTableEdit()

Endif

If (?InTable)
Message (&tbox;1000)
Result=MessageBox("Coach";&tbox;1) //OK Button

ELSE
Message (&tbox;1005) //No Table Created
IF(MessageBox("Coach";&tbox;4)<>1)

Break
Endif
Break

Endif
If (NOT Exists(Module))

RestoreStates()
Endif
RETURN
//*********************************************
// END MAIN PROGRAM
//*********************************************

//*********************************************
//
// SUBROUTINE: OverviewDlg
//
//*********************************************
Label(OverviewDlg)

Message(&tbox;1100) //Welcome Screen
Result=Welcome("Coach";&tbox)

If (Result=2)
RETURN

Endif
Return

//*********************************************
//
// SUBROUTINE: CreateTable
//
//*********************************************
Label(CreateTable)

//***********************************************************
// If already in a table, create new or edit this one?
//***********************************************************
While(?InTable)
If (?InTable)

Message(&tbox;1200) //Already a table
DlgCreate(Result; "Coach";DlgNoCancel!+DlgNoOk!; ;;62;8)

DlgControl (CtrlPushButton!;;"~Create New Table";;2;6)
DlgControl (CtrlPushButton!;;" ~Edit this Table";;22;6)
DlgControl (CtrlPushButton!;;"    ~Quit...Esc  ";StyCancel!;42;6)

FORNEXT (x;1;tbox[0])
DlgControl (CtrlLabel!;;tbox[x];;2;x+1;;)

Endfor
Dlgend

If (Result=2)
Return

Else  If(Result=-1)
QuitCoach()

Endif
Endif

Endif

Message(&tbox;1205) // Place cursor at table position
FakeIt("Coach";&tbox;3;1;0)

keypress={-8154;} // F7 key
chkcode={-1;}
Key=DoEdit(1;&chkcode;&keypress) //Wait for user to F7

Endwhile

WindowMaximize
Call(ValidMenu)
Message(&hbox;1215) //hint

Repeat
Message(&tbox;1210) //Choose Layout, Tables, Create
DisplayRewrite
Result=PostIt("Coach";&tbox;0;13;0;2;0)
MenuActive=0
Show=HotSpot(0;0;62;0;-537;&funkey;1)

Switch(Show) //Creating table
// *****************************************
// User chooses Show Me
//******************************************

CaseOf 1:
Result=Postit("Coach";&tbox;0;13;0;3;0)
ShowMenu("Coach";&sbox;52;2;3;5;1;0)

// *****************************************
// User chooses Layout 
//******************************************

CaseOf 1531: // User chooses Layout 

// *****************************************
// User presses Alt-L
//******************************************

CaseOf -537: // User pressed Alt-L
Result=Postit("Coach";&tbox;0;13;0;3;0)

 PullDownDlg
While(?Dialog=-1)
Endwhile
Type("l")

//********************************************************
// User Presses Alt+F7
//  Columns/Tables Dlg Box Dialog # 370
//********************************************************

CaseOf -8130: // User pressed Alt+F7

DisplayRewrite
DlgInput(on!)
ColumnsTablesDlg
TabKey Tabkey Tabkey
Call(TblColCreate)

// *****************************************
// User presses Alt-=
//******************************************

CaseOf -8085: // User pressed Alt+=
Message(&sbox;1220)
Result=PostIt("Coach";&tbox;0;13;0;3;0)
MenuActive=MenuBarControl("Coach"; &hbox1; &sbox; 52; 2; 108; &funkey)

// *****************************************
// No Choice
//******************************************

DEFAULT: // Default
MenuActive=MenuBarControl("Coach"; &hbox1; &sbox; 52; 2; 108; &funkey)

EndSwitch

If(MenuActive <> -1)
If (Show<>-8130) //User did not press Alt+F7

Result=MenuControl (&hbox1; &hbox2; &hbox3; &valid; &allow; &exallow; &accel; &exaccel; &subval;
&subacl; &exsubacl; &subs; &funkey)
DisplayRewrite
MenuActive=Result
If(Result=1538 or Result=99) //user selected Create or C

If(?Dialog<>-365)
CancelMenus()
CancelDialogs()
DisplayRewrite

TableCreateDlg
Endif

EndIf
Endif

Endif
Until(MenuActive<>-1)

//************************************************************
//  Table Create Dlg Box  Dialog #365
//************************************************************

If(?Dialog<>365) //if not at TableCreateDlg
CancelMenus()
TableCreateDlg

Endif

Result=0 Savescrn=0
//*************************************************************
Message(&tbox;1225)
Repeat

Result=PostIt("Coach";&tbox;12;16;0;1;0)
While(?Dialog = 365)

Look(Key)
If(key=-768)

If(?MouseLastY=yposS1)
If(?MouseLastX>=xposQ1 and ?MouseLastX<=xposQ2)

CancelMsg=0
CancelCoach()
key=0

EndIf
Endif

EndIf
If(key=-8160)

CancelMsg=0
CancelCoach()
key=0

EndIf
Type(ntoc(Key))

EndWhile
If(?LeftCode <> -12277)

DisplayRewrite
TableCreateDlg

Endif
Until(?LeftCode = -12277)
If(?Dialog<>36) //if not in table edit

TableEditDlg
Endif
RETURN

//*************************************************************

//***************************************************************
//
//  SUBROUTINE: Define Valid Menu items for Tables
//
//***************************************************************
LABEL (ValidMenu)

hbox1={"  Choose [Bold On]Layout.";▸
}

hbox2={"  Choose [Bold On]Tables.";▸
}

hbox3={"  Choose [Bold On]Create.";▸
}

LayoutMenu(&valid)
allow= { 1538;} // [Create]*
exallow= { 0;}
accel= { 116;} // T for Tables
exaccel={ 1;}
subval= { 1538; // Create *

1539; // Edit*
264; // Insert*
265; // Delete*
1540; // Join*
1541; // Split*
1542; // Calculate*
1543; // Create Floating Cell*
1544; // Edit Floating Cell*

}
subacl={ 99;} // c for (C)reate
exsubacl={0;

}
subs={1537;} // Tables*

funkey={-8130;} // Alt+F7
sbox={ "  Choose [Bold On]Create.";▸ //Message

"                       ";
}

HelpMsg={ "  Choose [Bold On]Tables,";▸
"[Bold Off]  then [Bold On]Create.";
"                  ";

}
RETURN

//************************************************************
//
// Routine for it they pressed Alt+F7 to Create
//
//************************************************************
Label (TblColCreate)

x1={19;20; //Tables, Create
}

x2={28;31;
}

y1={  7 ;8;
}

y2={   7; 8;
}

hotkey={ 116; //t tables
-768; //c create

}
hotpnt={0;1;}
helpmsg={ "  Choose [Bold On]Tables,";▸

"[Bold Off]  then [Bold On]Create.";
"                  ";

}

Result=0 SaveScrn=0

While(?Dialog = 370) //Columns/Tables Dlg Box
Message (&tbox;1230)
If(Result="Hint")

Message(&hbox;5010)
Moreinfo=MessageBox("Hint";&hbox;1)

Endif
Switch(?DialogControl)

CaseOf 0: //OK
info={ 0; //Control ID

370; //Dialog ID
0; //don't allow arrows
"Coach";        //Title
33; //xpos for Postit
6; //ypos for Postit
0; //Hint Button
"Coach"; //Help Title
1; //Restore Screen
0; //Reserved

}
accel={ 50; 116; //2 T for tables

}
aclpnt={0;0;}
Result=ButtonControl (&info; &x1; &x2; &y1; &y2; &hotkey; &hotpnt; &accel; &aclpnt; &tbox;
&helpmsg)

CaseOf 58: //Tables

info={ 58; //Control ID *
370; //Dialog ID
0; //don't allow arrows
"Coach";       //Title
33; //xpos for Postit
6; //ypos for Postit
0; //Hint Button
"Coach"; //help Title
1; //Restore Screen
0; //Reserved

}
accel={ 50;116; //2 or t for Table

}
aclpnt={0;0;}
Result=ButtonControl (&info; &x1; &x2; &y1; &y2; &hotkey; &hotpnt; &accel; &aclpnt; &tbox;
&helpmsg)

CaseOf 59: //Create
info={ 59; //Control ID *

370; //Dialog ID
0; //don't allow arrows
"Coach";        //Title
33; //xpos for Postit
6; //ypos for Postit
0; //Hint button
"Coach"; //Help Title
1; //Restore Screen
0; //Reserved

}
accel={ 49;99; //1 c Create

}
aclpnt={1;1}
Result=ButtonControl (&info; &x1; &x2; &y1; &y2; &hotkey; &hotpnt; &accel; &aclpnt; &tbox;
&helpmsg)

DEFAULT: //Default
info={ ?DialogControl; //Control ID

370; //Dialog ID
0; //don't allow arrows
"Coach";       //Title
33; //xpos for Postit
6; //ypos for Postit
0; //Hint Button
"Coach"; //Help Title
1; //Restore Screen
0; //Reserved

}
accel={ 50;116; // 2 t for Tables

}
aclpnt={0;0;}

allow={ 0;58;59;}
Result=InvalidControl (&info; &x1; &x2; &y1; &y2; &hotkey; &hotpnt; &accel; &aclpnt; &tbox;

&helpmsg; &allow)

EndSwitch
EndWhile

RETURN

//*********************************************
//
// PROCEDURE: Message
//
//*********************************************
PROCEDURE Message(&tbox;Msg)

Switch(Msg)

CaseOf 1000: //Step 1:  Successfully created Table
tbox={ "You have successfully created a table!";

" ";
}

CaseOf 1005: //No Table Created
tbox={ "Somehow no table was created in your document.";

"Would you like to try again?";
"";

}

CaseOf 1100: //About Tables--Overview

tbox={ " ";
"                      [Bold On]Create a Table[Bold Off]                     ";
" ";
"The Create a Table Coach will help you create or modify   ";
"a table.  A table can be used to organize your data into    ";
"columns and rows.";
" ";

           "╔═════════════════════════════════════════════════════════╗ ";
           "║                   [Bold On]Operating Expenses[Bold Off]                    ║";                         
           "╟───────────┬─────────────┬──────────────┬────────────────╢   ";
           "║ Expense   │Third Quarter│Fourth Quarter│ Difference     ║   ";
           "╟───────────┼─────────────┼──────────────┼───────────┬────╢   ";
           "║  Payroll  │ 289,800.00  │ 330,485.00   │ 40,685.00 │ 14%║   ";
           "╟───────────┼─────────────┼──────────────┼───────────┼────╢   ";
           "║  Taxes    │  12,075.00  │  35,500.00   │ 23,425.00 │194%║   ";
           "╟───────────┼─────────────┼──────────────┼───────────┼────╢   ";
           "║  Rent     │  29,600.00  │  29,600.00   │      0.00 │  0%║   ";

"╚═══════════╧═════════════╧══════════════╧═══════════╧════╝  ";
" ";                     

}

CaseOf 1200: //Already a table
tbox={ "Your cursor is already in a table. You must place";

"the cursor outside this table in order to create a new";
"table.  Would you like to...";
" ";

}

CaseOf 1205: //Place cursor where table should go

tbox={ "  Place the cursor where you want your table to appear.";▸
" ";

}

CaseOf 1210: //Choose Layout, Table, Create

tbox={
"To begin creating your table,";
" ";
"  Choose [Bold On]Layout[Bold Off],  ";▸
"  then [Bold On]Tables[Bold Off],";
"  then [Bold On]Create[Bold Off].";
"    or ";
"  Press [Bold On]Alt+F7[Bold Off].";▸
"";

}

CaseOf 1215:
hbox={ "  Choose [Bold On]Tables,";▸

"[Bold Off]  then [Bold On]Create.";
"                  ";

}

CaseOf 1220:
sbox={ "  Choose [Bold On]Tables,";▸

"[Bold Off]  then [Bold On]Create.";
"                   ";
}

CaseOf 1225:
tbox={ "  Enter the number of columns and rows you want.";▸

"  Choose [Bold On]OK[Bold Off].";▸
" ";
"Press [Bold On]Tab[Bold Off] to move around.";

" ";
}

CaseOf 1230:
tbox={ "  Choose [Bold On]Tables,";▸

"[Bold Off]  then [Bold On]Create.";
"                  ";

}

CaseOf 5000: //hint for Table Create Dlg

tbox= {
"Tables can be used to create and edit organized rows and";
"columns of data.  The area where a column and row intersect";
"is called a cell."; 
" ";
"Enter the number of columns and rows you would like your";
"table to contain.  A table can have up to 64 columns and";
"32,766 rows.";
" ";
"The area where a column and row intersect is called a cell."; 
" ";
"              Column A         Column B        Column C";
"         ╔════════════════╤════════════════╤═══════════════╗";

 "   Row 1 ║    Cell A1     │                │               ║";
"         ╟────────────────┼────────────────┼───────────────╢";

 "   Row 2 ║                │    Cell B2     │               ║";
 "         ╟────────────────┼────────────────┼───────────────╢";
 "   Row 3 ║                │                │   Cell C3     ║";
 "         ╚════════════════╧════════════════╧═══════════════╝";

" ";
}

CaseOf 5010: //hint for Columns/Tables Dlg Box
hbox={ "No hint available.  What type of information";

"would be helpful here?";
" ";

}

DEFAULT: //If none were found
tbox={ "Message not found!  ";

" ";
}

hbox={ "No hint available.  What type of information";
"would be helpful here?";
" ";

}

EndSwitch
ENDPROC

